
Introduction


One of the more exciting developments in the 3D arena is the popularization of VRML  (Virtual Reality Modelling Language) for the construction of 3D objects and 3D worlds. VRML viewers and editors are now becoming available permitting relatively inexperienced users to design their own objects and worlds. A secondary benefit is that VRML is quickly becoming the defacto standard format for translating between the many vendor specific object formats. In the past year or so, it was a selling point for a commercial editor to be able to read and write files in VRML format. Now, that selling point has changed into a requirement.





Any application dealing with 3D data as input must therefore be able to read VRML files. A prerequisite to implementing, say a 3D modeller or world editing tool, is a VRML reader. In this column, we will consider implementating such a reader.


How Much of VRML Should We Handle


VRML has gone through two major releases: VRML 1 which provided the basic facilities for describing hierarchically built 3D objects and VRML 2 which provided additional capabilities that permit programmed behaviors to be associated with the objects. If we wanted to build a VRML viewer that supported all the VRML 2 facilities, we would have to support 100% of the features and we would have to build a reader that could handle all of the VRML 2 syntax. Our goals are a bit more modest. We want to be able to take data produced by commerical modelers in VRML format and read it in. We are not concerned here with processing data after it is read(an editor, for example, might use this information to create corresponding objects whereas a translator might use it to generate output in a new format. Our goal is just to be able to read it in and have it available for subsequent processing.





A VRML file consists primarily of VRML nodes with the following syntax:





	NodeName {


		fieldName1 fieldValue1


		fieldName2 fieldValue2


		fieldName3 fieldValue3


			…


		fieldNamen fieldValuen


	}





Values can be (1) other nodes, (2) basic types like integers, floats, strings, or literal constants like TRUE or (3) even collections of such entries denoted by square brackets such as "[10 20 30]". Node names are easily distinguished from field names because only the former are capitalized. An example VRML file is shown in Listing 1.





But there are also some peculiarities. Items in a list can be optionally separated by commas. It is typical, for example, to see a list of vectors (each vector is one float) enumerated as follows:





	[10.1 10.2 10.3, 5.1 5.2 5.3, ]





Also, many modelers output trailing commas since they know the readers will ignore them. The converse also applies. There are nodes whose fields require vectors, for example, but no brackets are provided, as illustrated in the following:





	Inline {


		bboxSize 10 20.1 30


	}





VRML 1 has a special syntax for defining names (via the DEF keyword) and then subsequently reusing them (via the USE keyword); e.g., 





DEF LEG1 Transform {


	…


}


DEF LEG2 Transform {


	translation 0 0 1


	children [


		USE LEG1


	]


}





A DEF implies an automatic USE; LEG1 above is used twice; LEG2 is used only once.





VRML 1 permits certain special nodes like Group or Transform to contain an arbitrary number of subnodes without the existence of a field name but VRML 2 has eliminated this special feature. The LEG2 Transform, for example, has a field name called children. It is written in VRML 2 syntax. Field name children does not exist in VRML 1 syntax. 





VRML 2 introduces a facility for defining new node types (PROTO and EXTERNPROTO) where a user can describe the name, type, and initial value of the new field names along with a description of the new node's expansion in terms of previously known nodes. To associate new field names with the field names of nodes in the expansion, a special IS keyword is introduced. Another major feature introduced is an event-oriented framework for controlling the activity of the objects in a world. Event interconnections are specified via a syntax like





	ROUTE Node1.field1 TO Node2.field2





Our goal will be to handle all of VRML 1 and VRML 2 including DEF and USE but excluding PROTO, EXTERNPROTO, IS, and ROUTE. Additionally, VRML 1 nodes with subnodes that are not associated with any field name will be associated with the specific name “children”.


Designing the VRML Classes


Since VRML is so simple, we can get away with just two kinds of objects: VRMLNode and VRMLVariable. VRMLNode is essentially a dictionary variant since it maintains keys (field name) and values (other objects). It also keeps track of the parent permitting traversal both up and down the hierarchy. VRMLVariable is needed so that the application can control how variables are to be used. For example, for translation purposes, it may be adequate to have access to the name. For game building purposes, the application, by contrast, might which to make a shallow copy of the referenced object (or, alternatively, a deep copy). Consequently, our reader must not simply replace the use by the definition.





For grouping purposes, it is convenient to create a superclass VRMLObject. But the object that is our main focus is VRMLReader. It is essentially unrelated to the other VRML classes but it is convenient to place it in the same subhierarchy as shown in Figure 1.


The VRMLObject Class


Our first implementation of class VRMLObject simply provided a class method new that initializes and an instance method printOn: that redispatched message printOn:indentNext: to the receiver. The name of the method suggests that printing (of a node, for example) should start at the current stream position but that subsequent fields should be indented by the amount specified via the indentNext: keyword. Since VRML nodes can be nested arbitrarily deep, we just couldn’t decipher anything with a traditional printOn: method that didn’t indent. 





As our implementation evolved, even this was found to be inadequate. More specifically, fields that were collections had to be specially handled. For example, a collection of integers or floats could very well be handled with the standard printOn: (if we didn’t mind the extra parentheses) but a collection of VRML nodes couldn’t because the current indentation level would be lost. Moreover, each of these two possibilities could occur in two contexts: as a field in a VRML node and as the value of a VRML variable. 





So we added a method “privatePrint: object on: aStream indentNext: indent” (see Listing 2) that handles 4 cases: a normal object, a VRML object, a collection of VRML objects, and a collection of non-VRML objects (strings are handled like normal objects). It determines which of 4 selectors to use and “performs” it.


The VRMLNode and VRMLVariable Classes


Class VRMLObject (see Listing 3) keeps track of the node name, the fields, and its parent. The parent is set automatically whenever a field is set. Class VRMLVariable (see Listing 4) keeps track of the name and value (remember definitions are implied uses) and whether or not it is a definition (true for a definition, false for a use).


The VRMLReader Class


Class VRMLReader could be implemented using a traditional top-down recursive descent parser design. Indeed, that is what we would do if we were implementing the reader in a language like Java or C. But Smalltalk provides us with a simpler alternative. We first “filter” the input to create a suitable array literal and then we use the compiler to “compile” the literal. The resulting array is then recursively traversed to construct the appropriate VRML nodes and variables (a much simpler process since parsing is at the “word” level rather than the “character” level).





Class method read: (see Listing 5) creates an instance and sends it the message read:. In the instance method, we check the file header to determine whether or not we are reading a VRML 1 file. We use this information later to automatically construct “children” fields (where needed) for VRML 1 nodes. Method filteredString: provides the filtering mentioned above. It provides a number of different filtering needs: (1) it specifically discards VRML comments (C++ style comments) and spaces, (2) it prefixes floats that start with a period by character $0, (3) it replaces ‘0x’ by ‘16r’, and (4) it handles braces and square brackets by turning them into parentheses with a special header. For example, the string





Transform { #a box


children [


geometry Box {}


]


}





is filtered by changing it to





#(Transform (braces


children (squares


geometry Box (braces )


)))





Method processCollection: then traverses the resulting array (via a stream) and returns the corresponding collection of elements. Each element is picked up using nextItem: which determines what to do by performing the selector returned by whatsNext:. This selector is in the form “process…” corresponding either to one of the keywords DEF, USE, PROTO, EXTERNPROTO, IS, or ROUTE or to a field (lowercase name), node (uppercase name), or value (possibly a collection). Method processValue: is the more complex method since it deals with implicit collections of numbers or strings or explicit collections of nodes.


 


A number of support methods for performing parsing queries simplify the processing task; e.g., “isBraceList: item”, “isSquareList: item”, “isFieldName: name”, “isNodeName: name”, “isName: name”, “isNumberOrStringLiteral: item”.





We also made the following small addition to class collection.





	exists: aBooleanBlock


			"Answers true if aBooleanBlock evaluate to


			true for some element; false, otherwise."


		self detect: aBooleanBlock ifNone: [^false].


		^true 


Conclusions


Building a reader for VRML is not particularly difficult. The current reader handles all of the data representation aspects of VRML but doesn’t attempt to handle the specialized syntax needed for “full interactive” VRML 2. Perhaps some of you may wish to complete this implementation.


Where to Obtain the Code


Source code for this article can be obtained on the World Wide Web at http://www.objectpeople.com. It is implemented in ParkPlace/Digitalk’s Smalltalk Express but could easily be ported to other Smalltalk environments.





�





 Figure 1  The VRML hierarchy.








Listing 1 File sample.wrl.





#VRML V1.0 ascii





Separator {


	Info { string "Sample VRML file"}





	DirectionalLight {


		on         TRUE       # SFBool


		intensity  1          # SFFloat


		color      1 1 1      # SFColor


		direction  0 0 -1     # SFVec3f


	}





	ShapeHints {


		vertexOrdering	CLOCKWISE


		shapeType	SOLID


		}





	Transform {


		translation 0 0 0


		scaleFactor 1.0 1.0 1.0


		}





Separator { #Polyhedron


		DEF COORD99 Coordinate3 {


			point [


			-0.914400 1.524000 -24.384000 ,


			-0.914400 7.010400 -24.384000 ,


			-5.791200 7.010400 -24.384000 ,


			-11.887200 1.524000 -24.384000 ,


			-0.914400 1.524000 24.384000 ,


			-0.914400 7.010400 24.384000 ,


			-5.791200 7.010400 24.384000 ,


			-11.887200 1.524000 24.384000 ]


		} #Coordinate3





		Material {


			ambientColor [ 0.250000 0.250000 0.250000,


		               	0.250000 0.250000 0.250000,


		               	0.250000 0.250000 0.250000,


		               	0.250000 0.250000 0.250000,


		          		0.250000 0.250000 0.250000,


		      		0.250000 0.250000 0.250000,


				] #ambientColor


			diffuseColor [ 1.000000 1.000000 1.000000,


		               	1.000000 1.000000 1.000000,


		               	1.000000 1.000000 1.000000,


		               	1.000000 1.000000 1.000000,


		               	1.000000 1.000000 1.000000,


		               	1.000000 1.000000 1.000000,


				] #diffuseColor


			transparency [ 0.000000,


		               	0.000000,


		               	0.000000,


		               	0.000000,


		               	0.000000,


		               	0.000000,


				] #transparency


			} #Material





		Texture2 {


			filename [ "8tblr.bmp",


		               	"8tblr.bmp",


		               	"8tblr.bmp",


		               	"8tblr.bmp",


		               	"8tblr.bmp",


		               	"8tblr.bmp",


				] #filename


			} #Texture2





		IndexedFaceSet {


			coordIndex [ 0,1,2,3,-1,


		             		 0,3,7,4,-1,


		           		1,0,4,5,-1,


		       		2,1,5,6,-1,


		     		3,2,6,7,-1,


		     		7,6,5,4,-1,


				]


		} #IndexedFaceSet


} #Polyhedron


}








Listing 2 The VRMLObject class.





class:									VRMLObject


superclass:							Object


instance variables:			"none"





class methods 





instance creation


new


	^super new initialize 





instance methods





instance initialization


initialize


	"Default is to do nothing."





printing


printOn: aStream


	aStream cr.


	self privatePrint: self on: aStream indentNext: 3


   


private printing support


privatePrint: object on: aStream indentNext: indent


	self


		perform: (self privateObjectPrintingSelector: object)


		with: object


		with: aStream


		with: indent





privateObjectPrintingSelector: object


	(object isKindOf: VRMLObject)


		ifTrue: [^#privatePrintVRMLObject:on:indentNext:].


	object isCollection not | object isString


		ifTrue: [^#privatePrintBasicObject:on:indentNext:].


	(object exists: [:element | 


		(element isKindOf: VRMLObject) not])


		ifTrue: [^#privatePrintBasicObjects:on:indentNext:]


		ifFalse: [^#privatePrintVRMLObjects:on:indentNext:]





privatePrintBasicObject: object 


on: aStream indentNext: indent


	"This method is private to privatePrint:on:indentNext:."


	object printOn: aStream.





privatePrintBasicObjects: objects 


on: aStream  indentNext: indent


	"This method is private to privatePrint:on:indentNext:."


	aStream nextPutAll: ' ['.


	aStream cr; next: indent put: $ .


	objects do: [:object |


		object printOn: aStream. aStream space].


	aStream cr; next: indent - 3 put: $ ; nextPut: $].





privatePrintVRMLObject: object 


on: aStream indentNext: indent


	"This method is private to privatePrint:on:indentNext:."


	object printOn: aStream indentNext: indent


   


privatePrintVRMLObjects: objects 


on: aStream indentNext: indent


	"This method is private to privatePrint:on:indentNext:."


	aStream nextPutAll: ' ['.


	objects do: [:object |


		aStream cr; next: indent put: $ .


		object printOn: aStream indentNext: indent + 3].


	aStream cr; next: indent - 3 put: $ ; nextPut: $]. 








Listing 3 The VRMLNode class.





class:									VRMLNode 


superclass:							VRMLObject


instance variables:			name parent fields





instance methods





instance initialization


initialize


	name := 'Unknown'.


	fields := IdentityDictionary new





get/set methods


name


	^name


name: aSymbol


	name := aSymbol


 


parent


	^parent


parent: aVRMLObject


	parent := aVRMLObject





field accessing


at: fieldName


	^fields at: fieldName


   


at: fieldName ifAbsent: aBlock


	^fields at: fieldName ifAbsent: aBlock


 


at: fieldName put: anObject


	fields at: fieldName put: anObject.


	(anObject isKindOf: VRMLNode)


		ifTrue: [anObject parent: self]


   


printing support


printOn: aStream indentNext: indent


	| fieldName fieldValue |


	aStream nextPutAll: self name; space; nextPut: ${.


	fields associationsDo: [:association |


		fieldName := association key. 


		fieldValue := association value.


		aStream cr; next: indent put: $ .


		aStream nextPutAll: fieldName; nextPutAll: ' = '.


		self privatePrint: fieldValue 


			on: aStream indentNext: indent + 3].


	aStream cr; next: indent - 3 put: $ ; nextPut: $}. 








Listing 4 The VRMLVariable class.





class:									VRMLVariable


superclass:							VRMLObject


instance variables:			name value isDefinition





instance methods





get/set methods   


name


	^name


name: aSymbol


	name := aSymbol


   


value


	^value


value: aVRMLObject


	value := aVRMLObject 





isDefinition


	^isDefinition


isDefinition: aBoolean


	isDefinition := aBoolean





printing support


printOn: aStream indentNext: indent


	self isDefinition


		ifTrue: [


			aStream


				nextPutAll: 'DEF ';


				nextPutAll: self name;


				nextPutAll: ' = '.


			self privatePrint: self value 


				on: aStream indentNext: indent + 3]


		ifFalse: [


			aStream


				nextPutAll: 'USE ';


				nextPutAll: self name]








Listing 5 The VRMLReader class.





class:									VRMLReader


superclass:							VRMLObject


instance variables:			isVersion1 variables





class methods





examples


example1


	"self halt. VRMLReader example1"


	^self read: 'c:\wilf\junk.wrl'


 


reading


read: aPathName


	"Read the VRML file specified by aPathName."


	^self new read: aPathName 





instance methods





instance initialization


initialize


	variables := IdentityDictionary new





reading


read: aPathName


	"Read the VRML file specified by aPathName."


	| stream array |


	stream := File pathNameReadOnly: aPathName.


	(isVersion1 := (self isVersion1: stream)) |


	(self isVersion2: stream) ifFalse: [


		stream close.


		self error: 'This is not a VRML file'.


		^nil].





	array := self evaluate: (self filteredString: stream).


	stream close.





	array isNil ifTrue: [


		self error: 'Error in VRML file (see transcript)'.


		^nil].





	stream := ReadStream on: array.


	^self processCollection: stream





evaluation support


evaluate: aString


	"For VisualSmalltalk"


	 ^Compiler evaluate: aString in: UndefinedObject


		to: nil notifying: self ifFail: [nil]


	"For IBM Smalltalk


	^EsCompiler evaluate: aString for: nil


		ifFail: [:unused | nil]"


	"For VisualWorks


	^Compiler evaluate: aString"





compilerError: aString at: index in: aClass for: anObject


	"Do nothing"





filtering (special replacements)


filteredString: inputStream


	"Return a string containing an array literal


	with the following replacements:


		{...} by (braces ...)


		[...] by (squares ...)


		''...'' by '...'


		space .digit by space 0.digit


		0x... by 16r...


		, discarded


		comments discarded"





	| outputStream character |


	inputStream reset.


	outputStream := ReadWriteStream on: 


		(String new: inputStream size).


	outputStream nextPutAll: '#('.


	[inputStream atEnd ] whileFalse:[


		character := inputStream next.


		character == $, ifTrue: [


			outputStream space "discard"] ifFalse: [


		character == ${ ifTrue: [


			outputStream nextPutAll: '(braces '] ifFalse: [


		character == $} ifTrue: [


			outputStream nextPutAll: ')'] ifFalse: [


		character == $[ ifTrue: [


			outputStream nextPutAll: '(squares '] ifFalse: [


		character == $] ifTrue: [


			outputStream nextPutAll: ')'] ifFalse: [


		character == $" ifTrue: [


			outputStream 


				nextPut: $'; 


				nextPutAll: (inputStream upTo: $"); 


				nextPut: $'] ifFalse: [


		character == $# ifTrue: [


			inputStream nextLine] ifFalse: [


		character == $0 ifTrue: [


			(self isStreamX: inputStream)


				ifTrue: [


					inputStream next. 


					outputStream nextPutAll: '16r']


				ifFalse: [outputStream nextPut: $0]] ifFalse: [


		character isWhitespace ifTrue: [


			outputStream space.


			(self isStreamDotNumber: inputStream)


				ifTrue: [outputStream nextPut: $0]] ifFalse: [


		outputStream nextPut: character]]]]]]]]]].


	outputStream nextPutAll: ')'.


	^outputStream contents trimBlanks





item retrieval


nextItem: stream


	^self perform: (self whatsNext: stream peek) with: stream.





processing support


whatsNext: name


	name == #DEF ifTrue: [^#processDefinition:].


	name == #USE ifTrue: [^#processUse:].


	name == #PROTO ifTrue: [^#processProto:].


	name == #EXTERNPROTO ifTrue: [^#processExternProto:].


	name == #IS ifTrue: [^#processIs:].


	name == #ROUTE ifTrue: [^#processRoute:].


	(self isFieldName: name) ifTrue: [^#processField:].


	(self isNodeName: name) ifTrue: [^#processNode:].


	^#processValue: 





processing


processCollection: stream


	| collection item |


	collection := OrderedCollection new.


	[stream atEnd]  whileFalse: [


		(self isFieldName: stream peek) ifTrue: [


			self error: 'Field names ', stream next printString, 


				' not expected in a collection'].


		item := self nextItem: stream.


		item isCollection


			ifTrue: [collection addAll: item]


			ifFalse: [collection add: item]].


	^collection asArray


   


processDefinition: stream


	| name value |


	stream next.


	name := stream next.


	(variables includesKey: name) ifTrue: [


		^self error: 'DEF occurred for second time on ', 


			name printString].


	variables at: name put: (value := self nextItem: stream).


	^VRMLVariable new


		name: name;


		value: value;


		isDefinition: true


  


processUse: stream


	| name |


	stream next.


	name := stream next.


	^VRMLVariable new


		name: name;


		value: (variables at: name ifAbsent: [


			self error: 'USE name ', name printString, ' unknown']);


		isDefinition: false





processProto: nodeStream


	self error: 'prototype nodes not yet handled'


  


processExternProto: nodeStream


	self error: 'external prototype nodes not yet handled'


   


processIs: nodeStream


	self error: 'prototype IS not yet handled'





processRoute: nodeStream


	self error: 'event routing not yet handled'


 


processField: stream


	^Association


		key: stream next


		value: (self nextItem: stream)


  


processNode: nodeStream


	| node name array pair stream |


	"Prepare for new node."


	name := nodeStream next. array := nodeStream next.


	(self isBraceList: array) ifFalse: [


		self error: 'expected braces after node ', 


			node name printString].





	"Determine if we have a collection (which only 


	VRML1 supports) or named fields (which VRML1 


	and 2 support)."


	stream := ReadStream on: array.


	stream next. "get rid of brace"


	node := VRMLNode new name: name.


	(stream atEnd not and: [self isFieldName: stream peek])


		ifFalse: [


			node at: #children put: (self processCollection: stream).


			^node].





	"Process field names."


	[stream atEnd]  whileFalse: [


		pair := self nextItem: stream.


		pair isAssociation ifFalse: [


			self error: 'expected a field name in node ', name].


		node at: pair key put: pair value].


	^node





processValue: stream


	| data |


	data := stream next.


	(self isName: data) ifTrue: [^data].


	(self isNumberOrStringLiteral: data) ifTrue: [


		data := OrderedCollection with: data.


		[stream atEnd not and: [


		self isNumberOrStringLiteral: stream peek]] 


			whileTrue: [data add: stream next].


		data size = 1 ifTrue: [^data first] ifFalse: [^data asArray]].


	(self isBraceList: data) ifTrue: [


		self error: 'did not expect braces in this context'].


	(self isSquareList: data) ifTrue: [


		data := data copyFrom: 2 to: data size.


		^self processCollection: (ReadStream on: data)].


	self error: 'Unexpected data ', data printString





queries useful for parsing


isBraceList: item


	(item isKindOf: Array) ifFalse: [^false].


	item isEmpty ifTrue: [^false].


	^item first == #braces


 


isSquareList: item


	(item isKindOf: Array) ifFalse: [^false].


	item isEmpty ifTrue: [^false].


	^item first == #squares


   


isFieldName: name


	name isSymbol ifFalse: [^false].


	name isEmpty ifTrue: [^false].


	^name first isLowerCase


   


isNodeName: name


	name isSymbol ifFalse: [^false].


	name isEmpty ifTrue: [^false].


	^name first isUpperCase & (


		name last isLowerCase | name last isDigit)


   


isName: name


	name isSymbol ifFalse: [^false].


	name isEmpty ifTrue: [^false].


	^name first isLetter





isNumberOrStringLiteral: item


	item isNumber ifTrue: [^true].


	item isSymbol ifTrue: [^false].


	^item isString


 


isStreamDotNumber: stream


	| position isDigit |


	stream atEnd ifTrue: [^false].


	stream peek == $. ifFalse: [^false].


	position := stream position.


	stream next.


	stream atEnd ifTrue: [stream position: position. ^false].


	isDigit := stream peek isDigit.


	stream position: position.


	^isDigit





isStreamX: stream


	stream atEnd ifTrue: [^false].


	^stream peek asLowerCase == $x





isVersion1


	^isVersion1





isVersion1: stream


	^self streamHeader: stream matches: '#VRML V1.0 ascii'


 


isVersion2: stream


	^self streamHeader: stream matches: '#VRML V2.0 utf8'


   


streamHeader: stream matches: aString


	"Does the first line of the VRML file match the string?"


	stream size < aString size ifTrue: [^false].


	^(stream copyFrom: 1 to: aString size) = aString





printing    


printOn: aStream indentNext: indent


	aStream nextPutAll: 'aVRMLReader'

